

THE QUARTIC FORMULA: TI-84 PROGRAM

In the program listings which follow, the italic comments on the right provide hints for key strokes, and they are not part of the program. Other explanatory comments, also not part of the program, are preceded by a *.

One of the cases of the Quartic Formula is the $e = 0$ case, which requires solving a quadratic equation, which `prgmQUADRATI` does, *i.e.*, it solves $x^2 + px + q = 0$ for x . Note that the $+0i$ term under the radical tells the calculator that the result of the square root may be imaginary. In any case, before running `prgmQUADRATI`, store the coefficients of the polynomial in P and Q . Running the program then stores the two values of x in A and B .

```
prgmQUADRATI
:ClrHome                (PRGM I/O 8)
:√(P^2-4*Q+0i)→D        (STO>D)
:(1/2)*(-P+D)→A
:(1/2)*(-P-D)→B
:Disp "X [A,B]:" ,A,B   (PRGM I/O 3)
:Return                (PRGM CTL E)
:End                    (PRGM CTL 7)
```

The program `prgmQUARTIC` (listed on the next page) solves the quartic equation $x^4 + px^3 + qx^2 + rx + s = 0$ for x via the Quartic Formula. Again, note that the $+0i$ terms under the radicals tell the calculator that the results of the square roots may be imaginary (or complex). Before running `prgmQUARTIC`, store the coefficients of the polynomial in P , Q , R and S . Running the program then stores the values of x in H , I , J and K .

THE QUARTIC FORMULA: TI-84 PROGRAM

```

prgmQUARTIC
:ClrHome
:(1/8)*(8*Q-3*P^2)→D
:(1/8)*(8*R-4*P*Q+P^3)→E
:(1/256)*(256*S-64*P*R+16*P^2*Q-3*P^4)→F
+----:If E=0                                     (PRGM CTL 1 & 2ND TEST 1)
|      :Then                                     (PRGM CTL 2)
|      :P→Z                                     *This part is
|      :D→P                                     *the e = 0 case
|      :F→Q
|      :prgmQUADRATI                             (PRGM EXEC)
|      :√(A+0i)→C
|      :√(B+0i)→D
|      :(-1/4)*Z+C→H
|      :(-1/4)*Z-C→I
|      :(-1/4)*Z+D→J
|      :(-1/4)*Z-D→K
|      :Disp " "
|      :Disp "X [H,I,J,K]:",H,I,J,K
|      :Stop                                     (PRGM CTL F)
+----:Else                                     (PRGM CTL 3)
|      :(1/3456)*(2*D^3-72*D*F+27*E^2)→T         *a.k.a. D
|      :(1/442368)*(128*D^2*F^2+27*E^4-16*D^4*F
|      +4*D^3*E^2-144*D*E^2*F-256*F^3)→U         *a.k.a. E
|      :(1/144)*(D^2+12*F)→V                     *a.k.a. F
|      :(T+√(U+0i))^(1/3)→G
| +---:If V=0
| |    :Then
| |    :√(-D/6+G+0i)→A                           *for F = 0
| +---:Else
| |    :√(-D/6+G+V/G+0i)→A                         *for F ≠ 0
| +---:End
|      :√(A^2+D/2+E/4/A+0i)→B
|      :√(A^2+D/2-E/4/A+0i)→C
|      :(-1/4)*P+A+B*i→H                         *This part is
|      :(-1/4)*P+A-B*i→I                         *the e ≠ 0 case
|      :(-1/4)*P-A+C*i→J
|      :(-1/4)*P-A-C*i→K
|      :Disp "X [H,I,J,K]:",H,I,J,K
|      :Stop
+----:End
:End

```